```
#4.6 Question 7
> with(Groebner):
> with(PolynomialIdeals):

> I:=<x*z-y^2,z^3-x^5>;
```

$$I := \langle x\,z - y^2,\ z^3 - x^5 \rangle$$

```
#Part (a)
#We use the paratremizations in the hint we get
> J:=Basis( [x-t^3, y-t^4, z-t^5], `lexdeg`([t],[x,y,z])):
> J:=remove(has,J,t):
> J:=<op(J)>;
```

$$J := \langle y^2 - x\,z,\ -y\,z + x^3,\ -z^2 + x^2\,y \rangle$$

```
> K:=Basis( [x-t^3,y+t^4,z-t^5], `lexdeg`([t],[x,y,z])):
> K:=remove(has,K,t):
> K:=<op(K)>;
```

$$K := \langle y^2 - x\,z,\ y\,z + x^3,\ z^2 + x^2\,y \rangle$$

```
#Where I = J \intersect K We check this:
> Simplify(I);
```

$$\langle y^2 - x\,z,\ -z^3 + x^5 \rangle$$

```
> Simplify(Intersect(J,K));
```

$$\langle y^2 - x\,z,\ -z^3 + x^5 \rangle$$

```
>
#So we have that V(I) = V(J \intersect K) = V(J) \union V(K).
#Since V(J) = { (t^3,t^4,t^5), t in R } (it was constructed to do this)
#and similarly V(K) = { (t^3,-t^4,t^5) , t in R }, Prop 6 says that
#V(J) and V(K) are irreducible since they are definened parametrically.
#This means we have the desired decomposition into irreducible varieties.
>
#Part (b)
>
#4.5 Prop 3 gives that I(V(J)) and I(V(K)) are prime ideals sinve V(J), V(K)
#are irreducible.
>
#By our construction it is not hard to see I(V(J))=J and I(V(K))=K which mean
#J and K are also prime.
>
> IsRadical(J); IsRadical(K);
                                     true

                                     true

>
#Since the intersection of two radical ideal is radical we have that I is
#also radical.
>
> Quotient( I, K );
```

$$\langle y^2 - x\,z,\ -y\,z + x^3,\ -z^2 + x^2\,y \rangle$$

```
> Quotient( I, J );
```

$$\langle y^2 - x\,z,\ y\,z + x^3,\ z^2 + x^2\,y \rangle$$

```
>
#So we also have that I:J=K and I:K=J.
#We conclude that I = J \intersect K is the desired decomposition
#into primpe ideals that are ideal quotients of I.
>
#We verified that I = J \intersect K above.
```